
ECE 545—Digital System Design with VHDL
Lecture 8

Combinational Logic with Process Statements

VHDL Simulation: Signals vs. Variables

Midterm Review
10/21/08

Outline

• Behavioral VHDL Coding for Synthesis
• Coding for Combinational Logic using Process 

Statements
• Majority Function 
• Mux with don't cares
• Priority Encoder

• VHDL Timing: Signals vs. Variables

Resources

• Volnei A. Pedroni,Circuit Design with VHDL
• Chapter 6, Sequential Code

• Chapter 7, Signals and Variables

VHDL Design Styles (Architecture)

STRUCTURAL

components and
interconnects

VHDL Design 
Styles

DATAFLOW

“concurrent”
statements • Sequential Logic 

(registers,
counters)

• State machines 
(FSMs, ASMs)
• Complex Comb. 
Logic

“sequential” statements

NON-
SYNTHESIZABLE

SYNTHESIZABLE

BEHAVIORAL

• Test Benches
• Modeling IP• Gates

• Simple Comb. 
Logic

Coding for Combinational Logic

Behavioral Code for Combinational Logic

• Behavioral code (i.e. process statements) can be 
used to build combinational logic

• Up to this point, all combinational logic has been built 
using dataflow code (i.e. concurrent statements)
• Majority gate example

Majority Gate

3-input majority function

A B C F

0 0 0 0
0 0 1 0
0 1 0 0
0 1 1 1
1 0 0 0
1 0 1 1
1 1 0 1
1 1 1 1

• Logical expression form

F = A B + B C + A C

Majority Gate—Dataflow

library IEEE;

use IEEE.STD_LOGIC_1164.all;

entity majority1 is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

f : out STD_LOGIC

);

end majority1;

architecture dataflow of majority1 is

begin

f <= (a and b) or (a and c) or (b and c);

end dataflow;



9

Testbench

library ieee;

use ieee.std_logic_1164.all;

entity majority1_tb is

end majority1_tb;

architecture TB_ARCHITECTURE of majority1_tb is

component majority1

port(

a : in std_logic;

b : in std_logic;

c : in std_logic;

f : out std_logic );

end component;

signal a : std_logic := '0';

signal b : std_logic := '0';

signal c : std_logic := '0';

signal f : std_logic;

Testbench cont'd

begin

UUT : majority1 port map (a => a, b => b, c => c, f => f);

process

begin

for i in 0 to 1 loop

for j in 0 to 1 loop

for k in 0 to 1 loop  

wait for 10 ns;

c <= not c;

end loop;

b <= not b;

end loop;

a <= not a;

end loop;

wait;

end process;

end TB_ARCHITECTURE;

Waveform

Testbench—Another Method

library ieee;

use ieee.std_logic_1164.all;

use ieee.std_logic_unsigned.all;

entity majority1_tb is

end majority1_tb;

architecture TB_ARCHITECTURE of majority1_tb is

component majority1

port(

a : in std_logic;

b : in std_logic;

c : in std_logic;

f : out std_logic );

end component;

signal a : std_logic;

signal b : std_logic;

signal c : std_logic;

signal f : std_logic;

signal inputvec : std_logic_vector(2 downto 0);

Testbench—Another Method cont'd

begin

UUT : majority1 port map (a => a, b => b, c => c, f => f);

a <= inputvec(2); -- assign ports

b <= inputvec(1);

c <= inputvec(0);

process

begin

inputvec <= "000"; -- initialize

for i in 0 to 7 loop

wait for 10 ns;

inputvec <= inputvec + 1;

end loop;

wait;

end process;

end TB_ARCHITECTURE;

Combinational Logic Using Process Statements

• Combinational logic blocks can be designed using 
behavioral code

• When behavioral code mentioned, this means PROCESS 
statements

• Three important things to remember to prevent 
mistakes modeling combinational logic using process 
statements

1. Make sure all inputs (all things on right-hand side of an 
equation) are listed in process sensitivity list ��� � MISTAKE # 1

2. Make sure all signal outputshave a default assignment (or 
else may get a latch). Outputs should always get assigned 
some value for any input��� � MISTAKE #2

3. Remember that the statements in a process statement are 
sequential, so orders matters! ��� � MISTAKE #3

Examples illustrate mistakes

• The following examples will illustrate some common 
mistakes

• All examples use signals only, not variables
• Show correct example followed by mistakes

• Majority function � mistake #1, mistake #2

• Mux with don't cares � mistake #2

• Priority Encoder � mistake #3

16

Majority Function with Process Statements

Majority Function with Process Statements (Correct)
library IEEE;

use IEEE.STD_LOGIC_1164.all;

entity majority1 is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

f : out STD_LOGIC

);

end majority1;

architecture behavioral of majority1 is

begin

process (a,b,c)

begin

if (((a = '1') and (b = '1')) or ((a = '1') and (c = '1')) or 

((b = '1') and (c = '1')) ) then

f <= '1';

else

f <= '0';

end if;

end process;

end behavioral;

Mistake # 1—Forget input on sensitivity list
library IEEE;

use IEEE.STD_LOGIC_1164.all;

entity majority1 is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

f : out STD_LOGIC

);

end majority1;

architecture behavioral of majority1 is

begin

process (a,b) -- forget c on sensitivity list

begin

if (((a = '1') and (b = '1')) or ((a = '1') and (c = '1')) or 

((b = '1') and (c = '1')) ) then

f <= '1';

else

f <= '0';

end if;

end process;

end behavioral;

Waveform – Error # 1

Error: should be '1'

• Sometimes synthesis tool will find these
• Synplicity gives WARNINGS (in blue) to let us know it may have done something we 

did not intend it to do:
@W: CG296 … Incomplete sensitivity list - assuming co mpleteness
@W: CG290 … Referenced variable c is not in sensitiv ity list

Pre-Synthesis Simulation

Mistake # 2 – Not all outputs assigned value
library IEEE;

use IEEE.STD_LOGIC_1164.all;

entity majority1 is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

f : out STD_LOGIC

);

end majority1;

architecture behavioral of majority1 is

begin

process (a,b,c) 

begin

if (((a = '1') and (b = '1')) or ((a = '1') and (c = '1')) or 

((b = '1') and (c = '1')) ) then

f <= '1';

-- no default value assigning f <= '0'

end if;

end process;

end behavioral;



21

Waveform – Error # 2

• Synplicity gives WARNINGS (in blue) to let us know it may have done something we 
did not intend it to do:

@W: CL111 … All reachable assignments to f assign '1 ', register removed 
by optimization

@W: CL207 … All reachable assignments to f assign 1,  register removed by 
optimization

Error: should be '0'

Pre-Synthesis Simulation

Mux with Don't Care with Process Statements

Mux with Don't Cares

a

b

c

d

x

y

sel

2

d

c

b

a

x

-11

-10

101

000

ysel

'-' indicates don't care

Muxwith Don't Cares with Process Statements 
(Correct) 

library IEEE;

use IEEE.STD_LOGIC_1164.all; 

use IEEE.std_logic_arith.all;

use IEEE.std_logic_unsigned.all;

entity muxdontcare is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

d : in STD_LOGIC;

sel : in STD_LOGIC_VECTOR(1 downto 0);

x : out STD_LOGIC;

y : out STD_LOGIC

);

end muxdontcare;

Muxwith Don't Cares with Process Statements 
(Correct) cont'd

architecture behavioral of muxdontcare is

signal selint : integer range 0 to 3;

begin

selint <= conv_integer(unsigned(sel));

process (a,b,c,d,selint)

begin

if (selint = 0) then

x <= a;

y <= '0';

elsif (selint = 1) then

x <= b;

y <= '1';

elsif (selint = 2) then

x <= c;

y <= '-';  -- use '-' to indicate don't care

else -- selint is 3 

x <= d;

y <= '-';  -- use '-' to indicate don't care

end if;

end process;

end behavioral;

Testbench
library ieee;

use ieee.std_logic_unsigned.all;

use ieee.std_logic_arith.all;

use ieee.std_logic_1164.all;

entity muxdontcare_tb is

end muxdontcare_tb;

architecture TB_ARCHITECTURE of muxdontcare_tb is

component muxdontcare

port(

a : in std_logic;

b : in std_logic;

c : in std_logic;

d : in std_logic;

sel : in std_logic_vector(1 downto 0);

x : out std_logic;

y : out std_logic );

end component;

Testbench cont'd
signal a,b,c,d,x,y : std_logic;

signal sel : std_logic_vector(1 downto 0);

signal inputvec : std_logic_vector(3 downto 0);

begin

UUT : muxdontcare

port map ( a => a, b => b, c => c, d => d, sel => sel, x => x, y 
=> y );

a <= inputvec(3); b <= inputvec(2); c <= inputvec(1); d <= inputvec(0);

process

begin

inputvec <= "0000";

sel <= "00";

for i in 0 to 3 loop

for j in 0 to 15 loop

wait for 10 ns;

inputvec <= inputvec + 1;

end loop;

sel <= sel + 1;

end loop;

wait;

end process;

end TB_ARCHITECTURE;

Waveform

Pre-Synthesis Simulation

don't care



29

Waveform After Synthesis

When sel=10 and 11, synthesis tool will choose to put y to some value best suited for synthesis

i.e. y <= sel(0)

Post-Synthesis Simulation

Mistake # 2 – Not all outputs assigned value

library IEEE;

use IEEE.STD_LOGIC_1164.all; 

use IEEE.std_logic_arith.all;

use IEEE.std_logic_unsigned.all;

entity muxdontcare is

port(

a : in STD_LOGIC;

b : in STD_LOGIC;

c : in STD_LOGIC;

d : in STD_LOGIC;

sel : in STD_LOGIC_VECTOR(1 downto 0);

x : out STD_LOGIC;

y : out STD_LOGIC

);

end muxdontcare;

Mistake # 2 – Not all outputs assigned value cont'd

architecture behavioral of muxdontcare is

signal selint : integer range 0 to 3;

begin

selint <= conv_integer(unsigned(sel));

process (a,b,c,d,selint)

begin

if (selint = 0) then

x <= a;

y <= '0';

elsif (selint = 1) then

x <= b;

y <= '1';

elsif (selint = 2) then

x <= c;

-- don't assign y

else -- selint is 3 

x <= d;

-- don't assign y

end if;

end process;

end behavioral;

Waveform – Mistake #2

Error: creates a latch, i.e. remembers the last value of y

• Synplicity gives WARNINGS (in blue)to let us know it may have done something we 
did not intend it to do:

@W: CL117 … Latch generated from process for signal y, probably caused 
by a missing assignment in an if or case stmt

Pre-Synthesis Simulation

Inferred Latch

This latch should not be here, but was inferred due to poor coding

Priority Encoder with Process Statements

Signal Assignments in Processes

• In a process a signal assignment (<=) executes only 
once at the end of the process
• A variable executes immediately and can be assigned 

multiple times in a process

• If you do signal assignment multiple times (which is 
not recommended, except in certain cases), then the 
last signal assignmentchronologically will be 
assigned

Priority Encoder with Process Statements—Correct

library ieee;

use ieee.std_logic_1164.all;

entity priority is

port ( y: in std_logic_vector(7 downto 1);

pout: out std_logic_vector(2 downto 0) );

end priority;

architecture behavioral of priority is

begin

process (y)

begin

if (y(7) = '1') then pout <= "111";

elsif (y(6) = '1') then pout <= "110";

elsif (y(5) = '1') then pout <= "101";

elsif (y(4) = '1') then pout <= "100";

elsif (y(3) = '1') then pout <= "011";

elsif (y(2) = '1') then pout <= "010";

elsif (y(1) = '1') then pout <= "001";

else pout <= "000";

end if;

end process;

end behavioral;

NOTE: 
-for any possible input, only one 
"pout <=" statement assigned

Testbench
library ieee;

use ieee.std_logic_1164.all;

use ieee.std_logic_unsigned.all;

entity priority_tb is

end priority_tb;

architecture TB_ARCHITECTURE of priority_tb is

component priority

port(

y : in std_logic_vector(7 downto 1);

pout : out std_logic_vector(2 downto 0) );

end component;

signal y : std_logic_vector(7 downto 1);

signal pout : std_logic_vector(2 downto 0);

begin

UUT : priority

port map (

y => y,

pout => pout

);

Testbench cont'd
process

begin

y <= "0000000";

for i in 1 to 128 loop

wait for 10 ns;

y <= y + 1;

end loop;

wait;

end process;

end TB_ARCHITECTURE;

Waveform

Priority Encoder—Correct but be careful
library ieee;

use ieee.std_logic_1164.all;

entity priority is

port ( y: in std_logic_vector(7 downto 1);

pout: out std_logic_vector(2 downto 0) );

end priority;

architecture behavioral of priority is

begin

process (y)

begin

pout <= "000";

if (y(1) = '1') then pout <= "001"; end if;

if (y(2) = '1') then pout <= "010"; end if;

if (y(3) = '1') then pout <= "011"; end if;

if (y(4) = '1') then pout <= "100"; end if;

if (y(5) = '1') then pout <= "101"; end if;

if (y(6) = '1') then pout <= "110"; end if;

if (y(7) = '1') then pout <= "111"; end if;

end process;

end behavioral;

NOTES: 
-for any given input, "pout <=" statement 
assigned many times—this is not 
recommended, except in certain cases
-only the last "pout<=" assignment in the 
process counts
-therefore the highest priority assignment 
should be last!

highest priority last

Priority Encoder—Mistake #3 Sequence in Wrong 
Order

library ieee;

use ieee.std_logic_1164.all;

entity priority is

port ( y: in std_logic_vector(7 downto 1);

pout: out std_logic_vector(2 downto 0) );

end priority;

architecture behavioral of priority is

begin

process (y)

begin

pout <= "000";

if (y(7) = '1') then pout <= "111"; end if;

if (y(6) = '1') then pout <= "110"; end if;

if (y(5) = '1') then pout <= "101"; end if;

if (y(4) = '1') then pout <= "100"; end if;

if (y(3) = '1') then pout <= "011"; end if;

if (y(2) = '1') then pout <= "010"; end if;

if (y(1) = '1') then pout <= "001"; end if;

end process;

end behavioral;

NOTES: 
-for any given input, "pout <=" statement 
assigned many times—this is not 
recommend, except in certain cases
-only the last "pout<=" assignment 
counts
-this gives a priority encoder, but the 
wrong one: LSB has priority in this 
implementation!

highest priority last

Waveform – Mistake Shown

priority encoder behaves incorrectly!

Combinational Logic with Process Statements: Do 
Not Use Clocks!

• Combinational logic, whether coded in dataflow or 
behavioral logic, should not include clocksor resets

• In your coding, try to separate combinational logic 
from sequential logic
• Sequential logic should have process statements with 

clocks and resets

• Combinational logic should not have clocks or resets

• See Fall 2007 midterm solution, problem #4b, for 
example of separating sequential and combinational logic

VHDL Simulation: Signals vs. Variables

Main Point

• To avoid most problems with signals and variables
• 1) Do notuse variables (unless absolutely necessary)

• 2) Do notuse process statements for combinational logic 
(except when necessary/convenient, such as the output 
function and next state function in ASMs)

VHDL Simulation

• Event-Driven Simulation
• VHDL performs event-driven simulation

• When an event occurs at time tn, the VHDL simulator 
waits at this time, updates all pertinent signals/variables 
at that time, then moves to time tn+1

• This behavior can best be illustrated in a process 
statement

• Timing post-synthesis is easy to understand since 
there are physical gates involved

• Timing pre-synthesis is more difficult to understand

Event-Driven Simulation

Time tn

Events Events

Time tn+1

Events Events

Time tn+2

Events Events

Spacing between event times does not need to be uniform!

Based on: Zwolinski, "Digital System Design with VHDL"

Simulation with Different VHDL Coding Styles

• Dataflow/concurrent coding
• Output changes immediately when input changes
• Easy to understand!

• Structural coding
• Similar to dataflow coding
• Easy to understand!

• Process statement
• When one of inputs of the process statement changes, the process

statement "fires" at a certain time tn

• When process statement "fires" all signals/variables the process
uses are stored at that moment

• When process statement completes, the signals are updated 
(variables are updated in the process)

Example: Counter
library ieee ;

use ieee.std_logic_1164.all ;

use ieee.std_logic_unsigned.all ; 

entity upcount is

port ( reset, clock: in std_logic ;

q           : out std_logic_vector(3 downto 0) ) ;

end upcount ;

architecture behavior of upcount is

signal count : std_logic_vector(3 downto 0);

begin

upcount: process (clock)

begin

if (clock'event and clock = '1') then

if (reset = '1') then

count <= "0000" ;

else

count <= count + 1 ;

end if ;

end if;

end process;

q <= count;

end behavior ;

Process Statement Timing: Signals

architecture behavior of upcount is

signal count : std_logic_vector(3 downto 0);

begin

upcount: process (clock)

begin

if (clock'event and clock = '1') then

if (reset = '1') then

count <= "0000" ;

else

count <= count + 1 ;

end if ;

end if;

end process;

q <= count;

end behavior ;

1) when clock changes (i.e. clock goes high) 
store value of all signals/variables used in 
process at the exact moment clock changes

1

clock reset count

1 4

2) For signals in any if statements and on any 
right hand sideof equations, use these prior 
values

prior values

3) Store any signals on left hand sideas 
new values

5

count

new values

4) When leave the process, update any 
changed signals with new value

Process Statement Timing: Signals Cont'd

architecture behavior of upcount is

signal count : std_logic_vector(3 downto 0);

begin

upcount: process (clock)

begin

if (clock'event and clock = '1') then

if (reset = '1') then

count <= "0000" ;

else

count <= count + 1 ;

count <= count + 3;

end if ;

end if;

end process;

q <= count;

end behavior ;

1

clock reset count

1 4

NOTE 1: For signals, right-hand side signal 
always uses the prior value

NOTE 2: If there are two or more 
assignments to the same signal, the LAST 
assignment to be executed determines the 
new value.

prior values

7

count

new values

library ieee;

use ieee.std_logic_1164.all;

entity priority is

port ( y: in std_logic_vector(7 downto 1);

pout: out std_logic_vector(2 downto 0) );

end priority;

architecture behavioral of priority is

begin

process (y)

begin

pout <= "000";

if (y(7) = '1') then pout <= "111"; end if;

if (y(6) = '1') then pout <= "110"; end if;

if (y(5) = '1') then pout <= "101"; end if;

if (y(4) = '1') then pout <= "100"; end if;

if (y(3) = '1') then pout <= "011"; end if;

if (y(2) = '1') then pout <= "010"; end if;

if (y(1) = '1') then pout <= "001"; end if;

end process;

end behavioral;
001

NOTE 1: only LAST 
signal assignment 
executed will 
determine the 
updated value of pout

NOTE 2:  pout always
has a new value and 
neverdepends on 
previous value of 
pout. � This tells 
VHDL that this is a 
combinational logic 
block (i.e. no flip-
flops)!

some
value

prior value

new value

Revisiting the Priority Encoder using Signals

pout

pout

Process Statement: Signals in Testbenches

• The main exception to the above rules for signals is 
in a testbench when a process statement does not 
have a sensitivity list

• In testbenches, the process of storing, computing, 
and updating signals occurs before/after a WAIT
command (or other similar commands)

Testbench Timing

process

begin

a <= "0000";

c <= '0';

b <= "1001"; -- set b to a constant value

wait until (clock'event and clock='0'); 
reset <= '1';

wait for clockperiod;

reset <= '0';

a <= a + 1;

a <= a + 2;

wait for clockperiod;

. . .

store prior values

update new values
(signal is updated only once,
according to a + 2)

Process Statement Timing: Signals vs. Variables

library ieee ;

use ieee.std_logic_1164.all ;

use ieee.std_logic_unsigned.all ; 

entity upcount_var is

port ( reset, clock: in std_logic ;

q,qvar : out std_logic_vector(3 downto 0) ) ;

end upcount_var ;

architecture behavior of upcount_var is

signal count : std_logic_vector(3 downto 0);

Process Statement Timing: Signals vs. Variables

begin

upcount: process (clock)

variable countvar : std_logic_vector(3 downto 0);

begin

if (clock'event and clock = '1') then

if (reset = '1') then

count <= "0000" ; 

countvar := "0000";

else

count <= count + 2 ;  

countvar := countvar + 2;

count <= count + 1 ;

countvar := countvar + 1;

end if ;

end if;

qvar <= countvar;

end process;

q <= count;

end behavior ;

NOTE 1: Variables update instantly, 
not only at the end of a process. So 
countvar becomes 2, then 3. Signal 
only updates once, becoming 1.

NOTE 2: Variables are local only to a 
process. To get information about a 
variable outside of the process, must 
assign variable to signal before process 
ends.

prior values (signals/variables)

count

0

countvar

0

new values (signals/variables)

count

1

countvar

3

Process Statement Timing: Signals vs. Variables 
Simulation

Notice that qvar increments by 3, q increments by 1.

Another downside of using variables is that, due to their instantaneous update,
they are usually not visible in simulation. This becomes a major problem with DEBUG.
Signals are always visible in simulation (you can open your hierarchy and pull up any
signal).

58

Midterm Review

Midterm Specifics

• The midterm will take place in class on Tuesday, October 28, 
during the class period from 4:30 pm – 7:10 pm.

• The midterm will be open-book, open-notes.
• You can bring any textbooks
• No electronic devices (cell phones, PDAs, laptops, etc.) are allowed 

� you must PRINT OUTall notes on paper. 
• Do not come to class with your notes in electronic form on your 

laptop, as you will not be allowed to use your laptop.
• The midterm will cover all material from lectures, homeworks, 

projects, hands-on sessions, and textbook readings.
• Exception: there will be no problems in which you need to use the 

FPGA CAD tools.
• We will go over last year's midterm in class today to give you a

better feel for example midterm problems.

Midterm Specifics Cont'd

• TIP: There will be a problem on algorithmic state machines 
on the exam. To help with this problem, you can study:
• Bit-Counter Example (lecture)
• Sorting Example (lecture/homework)
• Log2 Example (homework)
• Project Hash Function Example (project)
• 2007 ECE 545 Midterm (lecture)
• Session 3 Hands-On Session Example (posted on website)

• This example shows use of file I/O which we have not learned yet and which 
will not be tested on the midterm.

Complex Digital System Design with ASMs: 
Overview

CONTROLLER DATAPATH

control signals

status signals

entity inputs

entity outputs

entity inputs

entity output

mydesign.vhd

datapath.vhdcontroller.vhd

• Top-level design composed of two components: 
datapath and controller

Datapath

DATAPATH
datapath.vhd

• Datapath composed of various combinational logic and sequential logic building blocks
• Inputs:

• External inputs (Datain, etc.)
• Control signals from controller (enable, load, etc.)

• Outputs:
• External outputs (DataOut, etc.)
• Status signals to controller (counter reached a certain value, comparator output, etc.)

external inputs

external outputs

control

status

Controller

CONTROLLER
controller.vhd

Csel 0 = Int 1 = Ain, , 

Csel 0 = Int 1 = Wr Bout, , , 

Csel 1 = Int 1 = Wr Aout, , , 

Bin Csel 1 = Int 1 = , , 

• Controller is a hybrid Mealy/Moore finite state machine (drawn as a detailed controller ASM)
• Inputs (diamonds in the ASM)

• External inputs (Reset, etc.)
• Status signals from the datapath (counter reached a certain value, comparator output, etc.) 

• Outputs (ovals or rectangles):
• External outputs (Done, etc.)
• Control signals to datapath (enable, load, etc.)

control

status

external inputs

external outputs

Controller Implemented in Hardware

Combinational
Logic

Combinational
Logic

Csel 0 = Int 1 = Ain, , 

Csel 0 = Int 1 = Wr Bout, , , 

Csel 1 = Int 1 = Wr Aout, , , 

Bin Csel 1 = Int 1 = , , 

=
present state

external outputs,
control outputs
(to datapath)

external inputs,
status inputs

(from datapath)

next state

clock reset

Top-Level Entity

Combinational
Logic

Combinational
Logic

clock reset

Combinational
and

Sequential Logic

Combinational
and

Sequential Logic

external inputs external inputs

external outputs

external outputs

controller.vhddatapath.vhd

status

control

mydesign.vhd

Midterm Review

• We will go through last year's midterm in class


